Universidad del País Vasco — Euskal Herriko Unibertsitatea

Autoevaluación

OCW 2020: *Parametrización y representación gráfica de superficies construidas*

Test nº2 (resolución)

Equipo docente del curso
Martín Yagüe, Luis
Barrallo Calonge, Javier
Soto Merino, Juan Carlos
Lecubarri Alonso, Inmaculada

Departamento de Matemática Aplicada
Escuela de Ingeniería de Bilbao, Edificio II-I (EIB/BIE)
ETS de Arquitectura de Donostia-San Sebastián (ETSASS/DAGET)

EJERCICIOS DE AUTOEVALUACIÓN: Test nº2

Ejercicio nº1

Enunciado

Determine los valores del parámetro $t \in \mathbb{R}$ para los que intersectan el plano $y = 1$ y la circunferencia:

$$C(t) = \left(1 + \sqrt{2}\ \cos(t),\ \sqrt{2}\ \mathrm{sen}(t),\ 0\right) \quad \forall\, t \in [0, 2\pi]$$

Resolución

```
Remove["Global`*"] (* Borrado de variables utilizadas *)
```

- definición de ecuaciones paramétricas de la circunferencia

```
circ = {x[t_] = 1 + √2 Cos[t], y[t_] = √2 Sin[t], z[t_] = 0};
```

- cálculo de $t \in [0, 2\pi]$ cuando $y = 1$

```
Solve[y[t] == 1, t]
```

$$\left\{\left\{t \to \text{ConditionalExpression}\left[\frac{\pi}{4} + 2\pi\, C[1],\ C[1] \in \mathbb{Z}\right]\right\},\ \left\{t \to \text{ConditionalExpression}\left[\frac{3\pi}{4} + 2\pi\, C[1],\ C[1] \in \mathbb{Z}\right]\right\}\right\}$$

- infinitas soluciones (dos soluciones cada $2\pi\, rad$) pero sólo válidas $t_1 = \frac{\pi}{4}$, $t_2 = \frac{3\pi}{4} \in [0, 2\pi]$
- representación gráfica

```
circ1 = ParametricPlot3D[circ, {t, 0, 2 Pi}];

planoy1 = ContourPlot3D[y == 1, {x, -0.5, 2.5}, {y, -1, 1}, {z, -1, 1},
   Mesh → False, ContourStyle → {LightGreen, Opacity[0.4]}, AspectRatio → Automatic];

puntos = {circ /. t → Pi / 4, circ /. t → 3 Pi / 4};

p1 = ListPointPlot3D[puntos, PlotStyle → {Red, PointSize[Large]}];

Show[circ1, planoy1, p1, PlotRange → All]
```



- Solución: $\boxed{\text{Opción } \boldsymbol{d}}$

Ejercicio nº2

Enunciado

Calcule el punto de la circunferencia $C(t) = \left(1 + \sqrt{2}\,\cos(t),\ \sqrt{2}\,\text{sen}(t),\ 0\right)\ \ \forall\, t \in [0, 2\pi]$ correspondiente al valor del parámetro $t = \frac{4\pi}{9}$

Resolución

```
Remove["Global`*"]
```

- definición de ecuaciones paramétricas de la circunferencia

```
circ = {x[t_] = 1 + √2 Cos[t], y[t_] = √2 Sin[t], z[t_] = 0};
```

- cálculo de coordenadas cartesianas del punto (cuando $t = \frac{4\pi}{9}$)

```
puntos = {circ /. t → 4 Pi / 9} // N
{{1.24558, 1.39273, 0.}}
```

- punto: $P = (1.24558, 1.39273)$
- Solución: $\boxed{\text{Opción } \boldsymbol{d}}$

Ejercicio nº3

Enunciado

Determine el valor del parámetro $t \in \mathbb{R}$ para el que se obtiene el punto $A = (-4.12, 0.96, 4.12)$ de la recta que pasa por los puntos $P = (-1, 2, 1)$ y $Q = (2, 3, -2)$.

Resolución

```
Remove["Global`*"]
```

- definición de puntos y vector director

```
puntos = {p = {-1, 2, 1}, q = {2, 3, -2}}; vr = q - p;
```

- parametrización

```
r = {xr = p[[1]] + vr[[1]] t, yr = p[[2]] + vr[[2]] t, zr = p[[3]] + vr[[3]] t}
{-1 + 3 t, 2 + t, 1 - 3 t}
```

$$r \equiv \begin{cases} x(t) = & -1 + 3t \\ y(t) = & 2 + t \\ z(t) = & 1 - 3t \end{cases} \quad \forall\, t \in \mathbb{R}$$

- cálculo del parámetro

```
pa = {-4.12, 0.96, 4.12};

Solve[r == pa, t]
{{t → -1.04}}
```

- solución única: $t = -1.04$
- representación gráfica

```
punt = ListPointPlot3D[{puntos, {pa}}, PlotStyle → {Blue, Red, PointSize[Thickness[Large]]}];

recta = ParametricPlot3D[r, {t, -2, 2.2}, AxesLabel → {x, y}, PlotStyle → {Green, Thick}];

Show[punt, recta, PlotRange → All, AspectRatio → Automatic, AxesLabel → {X, Y, Z}]
```



- Solución: $\boxed{\text{Opción } \boldsymbol{d}}$

Ejercicio nº4

Enunciado

Indique la parametrización correcta de la cónica: $C \equiv \begin{cases} 9x^2 + 4y^2 + 18x - 16y - 11 = 0 \\ z + 1 = 0 \end{cases}$

Resolución

```
Remove["Global`*"]
```

- se trata de una elipse en el plano $z = -1$

```
m = CoefficientList[9 x^2 + 4 y^2 + 18 x - 16 y - 11, {x, y}]
```

```
{{-11, -16, 4}, {18, 0, 0}, {9, 0, 0}}
```

- coordenadas del centro

```
centro = {-(m[[2, 1]] / 2)/(m[[3, 1]]), -(m[[1, 2]] / 2)/(m[[1, 3]])}
```

$$\text{centro} = \left\{-\frac{m[[2,1]]/2}{m[[3,1]]}, -\frac{m[[1,2]]/2}{m[[1,3]]}\right\}$$

```
{-1, 2}
```

- valor de K

$$k = \frac{(m[[2,1]]/2)^{\wedge}2}{m[[3,1]]} + \frac{(m[[1,2]]/2)^{\wedge}2}{m[[1,3]]} - m[[1,1]]$$

```
36
```

- semiejes/radio

$$\text{If}\left[m[[3,1]] == m[[1,3]],\ \text{radio} = \sqrt{\frac{k}{m[[3,1]]}},\ \text{semiejes} = \left\{\sqrt{\frac{k}{m[[3,1]]}}, \sqrt{\frac{k}{m[[1,3]]}}\right\}\right]$$

```
{2, 3}
```

- ecuación canónica: $C \equiv \begin{cases} \frac{(x+1)^2}{4} + \frac{(y-2)^2}{9} - 1 = & 0 \\ z+1 = & 0 \end{cases}$

- ecuaciones paramétricas: $C \equiv \begin{cases} x(t) = & -1+2\cos(t) \\ y(t) = & 2+3\,\text{sen}(t) \\ z(t) = & -1 \end{cases} \quad t \in [0, 2\pi]$

- definición de ecuaciones paramétricas y representación gráfica

```
elip = {x[t_] = -1 + 2 Cos[t], y[t_] = 2 + 3 Sin[t], z[t_] = -1};

elip4 = ParametricPlot3D[elip, {t, 0, 2 Pi}];

Show[elip4, AspectRatio → Automatic, AxesLabel → {X, Y, Z}]
```



- Solución: $\boxed{\text{Opción } \boldsymbol{d}}$

Ejercicio nº5

Enunciado

Indique la parametrización correcta de la cónica: $C \equiv \begin{cases} 2x^2 - \sqrt{2}\, z - 1 = & 0 \\ x - y = & 0 \end{cases}$

Resolución

```
Remove["Global`*"]
```

- se trata de una parábola contenida en el plano $y = x$
- parametrización

```
par = {xp[t_] = t, yp[t_] = xp[t], zp[t_] = (2 xp[t]^2 - 1) / √2}
```

$\left\{t, t, \frac{-1+2t^2}{\sqrt{2}}\right\}$

$$C \equiv \begin{cases} x(t) = & t \\ y(t) = & t \\ z(t) = & \frac{2t^2-1}{\sqrt{2}} \end{cases} \quad \forall\, t \in \mathbb{R}$$

- representación gráfica

```
par5 = ParametricPlot3D[par, {t, -1, 1}];
```

```
plano5 = ContourPlot3D[y == x, {x, -1, 1}, {y, -1, 1}, {z, -0.7, 0.7},
   Mesh → False, ContourStyle → {LightGreen, Opacity[0.4]}, AspectRatio → Automatic];
```

```
Show[par5, plano5, PlotRange → All, AspectRatio → Automatic, AxesLabel → {X, Y, Z}]
```



- Solución: $\boxed{\text{Opción } \boldsymbol{b}}$

Ejercicio nº6

Enunciado

Determine la recta tangente en el punto $P = (0, 1, 0)$ a la curva: $\begin{cases} x(t) = & \cos(t) \\ y(t) = & \operatorname{sen}(t) \\ z(t) = & \operatorname{sen}(2t) \end{cases}$ $t \in [0, 2\pi]$

Resolución

```
Remove["Global`*"]
```

- parametrización

```
c6 = {x[t_] = Cos[t], y[t_] = Sin[t], z[t_] = Sin[2 t]}
```

```
{Cos[t], Sin[t], Sin[2 t]}
```

- cálculo del valor del parámetro $t \in [0, 2\pi]$ en el punto $P = (0, 1, 0)$

```
p = {0, 1, 0};
```

```
Solve[{x[t] == p[[1]], y[t] == p[[2]], z[t] == p[[3]]}, t]
```

$\left\{\left\{t \to \text{ConditionalExpression}\left[\frac{\pi}{2} + 2\pi\, C[1],\ C[1] \in \mathbb{Z}\right]\right\}\right\}$

- vector director de la recta tangente en el punto $P = (0, 1, 0)$

```
v6 = D[c6, t] /. t → Pi / 2
```

```
{-1, 0, -2}
```

- parametrización

```
t6 = {xt[t_] = p[[1]] + v6[[1]] t, yt[t_] = p[[2]] + v6[[2]] t, zt[t_] = p[[3]] + v6[[3]] t}
```

```
{-t, 1, -2 t}
```

$$r \equiv \begin{cases} x(t) = & -t \\ y(t) = & 1 \\ z(t) = & -2t \end{cases} \quad \forall\, t \in \mathbb{R}$$

- representación gráfica

```
cur6 = ParametricPlot3D[c6, {t, 0, 2 Pi}, PlotStyle → {Blue, Thickness[0.025]}];

r6 = ParametricPlot3D[t6, {t, -1, 1}, PlotStyle → {Green, Thickness[0.015]}];

punt = ListPointPlot3D[{p}, PlotStyle → {Red, PointSize[Large]}];

Show[cur6, r6, punt, PlotRange → All, AspectRatio → Automatic, AxesLabel → {X, Y, Z}]
```



- Solución: $\boxed{\text{Opción } \boldsymbol{b}}$

Ejercicio nº7

Enunciado

Indique el plano en el que está contenida la siguiente curva:

$$C(t) = \left(1 + \frac{\cos(t)}{\sqrt{3}} + \frac{\operatorname{sen}(t)}{\sqrt{2}},\ 1 - \frac{\cos(t)}{\sqrt{3}} + \frac{\operatorname{sen}(t)}{\sqrt{2}},\ \frac{\operatorname{sen}(t)}{\sqrt{2}}\right) \quad \forall\, t \in [0, 2\pi]$$

Resolución

```
Remove["Global`*"]
```

- parametrización

$$\texttt{c7} = \left\{\texttt{x7[t\_]} = 1 + \frac{\texttt{Cos[t]}}{\sqrt{3}} + \frac{\texttt{Sin[t]}}{\sqrt{2}},\ \texttt{y7[t\_]} = 1 - \frac{\texttt{Cos[t]}}{\sqrt{3}} + \frac{\texttt{Sin[t]}}{\sqrt{2}},\ \texttt{z7[t\_]} = \frac{\texttt{Sin[t]}}{\sqrt{2}}\right\}$$

$$\left\{1 + \frac{\texttt{Cos[t]}}{\sqrt{3}} + \frac{\texttt{Sin[t]}}{\sqrt{2}},\ 1 - \frac{\texttt{Cos[t]}}{\sqrt{3}} + \frac{\texttt{Sin[t]}}{\sqrt{2}},\ \frac{\texttt{Sin[t]}}{\sqrt{2}}\right\}$$

- sustituyendo $x(t)$, $y(t)$, $z(t)$ en la ecuación $Ax + By + Cz + D = 0$

```
ec7 = a*x7[t] + b*y7[t] + c*z7[t] + d == 0 // Simplify
```

$$a + b + d + \frac{(a-b)\,\text{Cos}[t]}{\sqrt{3}} + \frac{(a+b+c)\,\text{Sin}[t]}{\sqrt{2}} == 0$$

- la ecuación anterior debe verificarse $\forall\, t \in [0, 2\pi]$

```
s7 = Solve[{a + b + d == 0, a - b == 0, a + b + c == 0}, {a, b, c, d}]
```

Solve: Equations may not give solutions for all "solve" variables.

```
{{b → a, c → -2 a, d → -2 a}}
```

- ecuación del plano: $\pi \equiv x + y - 2z - 2 = 0$

```
a*x + b*y + c*z + d == 0 /. s7[[1]] // Simplify
```

```
a (x + y - 2 (1 + z)) == 0
```

- representación gráfica

```
g7 = ParametricPlot3D[c7, {t, 0, 2 Pi}];

plano7 = ContourPlot3D[x + y - 2 z == 2, {x, 0, 2}, {y, 0, 2}, {z, -3, 1},
    Mesh → False, ContourStyle → {LightGreen, Opacity[0.4]}, AspectRatio → Automatic];

Show[g7, plano7, PlotRange → All, AspectRatio → Automatic, AxesLabel → {X, Y, Z}]
```



- Solución: $\boxed{\text{Opción } e}$

Ejercicio nº8

Enunciado

Obtenga la forma implícita de la siguiente curva:

$$C(t) = \left(1 + \frac{\cos(t)}{\sqrt{3}} + \frac{\text{sen}(t)}{\sqrt{2}},\ 1 - \frac{\cos(t)}{\sqrt{3}} + \frac{\text{sen}(t)}{\sqrt{2}},\ \frac{\text{sen}(t)}{\sqrt{2}}\right) \quad \forall\, t \in [0, 2\pi]$$

Resolución

- eliminación del parámetro

```
Eliminate[{x == 1 + Cos[t]/Sqrt[3] + Sin[t]/Sqrt[2], y == 1 - Cos[t]/Sqrt[3] + Sin[t]/Sqrt[2], z == Sin[t]/Sqrt[2]}, t]
```

··· Eliminate: Inverse functions are being used by Eliminate, so some solutions may not be found; use Reduce for complete solution information.

```
x == 2 - y + 2 z && 3 y^2 + y (-6 - 6 z) == -2 - 6 z - 5 z^2
```

```
Factor[3 y^2 + y (-6 - 6 z) == -2 - 6 z - 5 z^2] // Simplify
```

```
2 + 3 y^2 + 6 z + 5 z^2 == 6 y (1 + z)
```

- la forma implícita resultante es $C \equiv \begin{cases} 3\,y^2+5\,z^2-6\,yz-6\,y+6\,z+2= & 0 \\ x+y-2\,z-2= & 0 \end{cases}$

- agrupando términos $C \equiv \begin{cases} 3\,(y-1)^2+5\left(z+\frac{3}{5}\right)^2-6\,yz+\frac{14}{5}= & 0 \\ x+y-2\,z-2= & 0 \end{cases}$

- representación gráfica: intersección entre cilindro y plano

```
c8 = ContourPlot3D[3 y^2 + y (-6 - 6 z) == -2 - 6 z - 5 z^2, {x, 0, 2}, {y, 0, 2}, {z, -3, 1},
   Mesh → False, ContourStyle → {LightGreen, Opacity[0.4]}, AspectRatio → Automatic];
```

```
plano8 = ContourPlot3D[x + y - 2 z == 2, {x, 0, 2}, {y, 0, 2}, {z, -3, 1},
   Mesh → False, ContourStyle → {LightRed, Opacity[0.4]}, AspectRatio → Automatic];
```

  - se usa la variable **g7** del ejercicio anterior

```
Show[c8, plano8, g7, PlotRange → All, AspectRatio → Automatic, AxesLabel → {X, Y, Z}]
```



- Solución: $\boxed{\text{Opción } \boldsymbol{b}}$

Ejercicio nº9

Enunciado

Calcule la proyección sobre el plano $x = 0$ de la curva:

$$C\,(t) = \left(1+\frac{\cos\,(t)}{\sqrt{3}}+\frac{\operatorname{sen}\,(t)}{\sqrt{2}},\; 1-\frac{\cos\,(t)}{\sqrt{3}}+\frac{\operatorname{sen}\,(t)}{\sqrt{2}},\; \frac{\operatorname{sen}\,(t)}{\sqrt{2}}\right) \quad \forall\; t\in[0,\,2\,\pi]$$

Resolución

- eliminación del parámetro t (hecho en el ejercicio anterior)

```
Eliminate[{x == 1 + Cos[t]/Sqrt[3] + Sin[t]/Sqrt[2], y == 1 - Cos[t]/Sqrt[3] + Sin[t]/Sqrt[2], z == Sin[t]/Sqrt[2]}, t]
```

```
x == 2 - y + 2 z && 3 y^2 + y (-6 - 6 z) == -2 - 6 z - 5 z^2
```

```
Factor[3 y^2 + y (-6 - 6 z) == -2 - 6 z - 5 z^2] // Simplify
```

```
2 + 3 y^2 + 6 z + 5 z^2 == 6 y (1 + z)
```

- la forma implícita resultante es $C \equiv \begin{cases} 3y^2 + 5z^2 - 6yz - 6y + 6z + 2 = 0 \\ x + y - 2z - 2 = 0 \end{cases}$

- agrupando términos $C \equiv \begin{cases} 3(y-1)^2 + 5\left(z + \frac{3}{5}\right)^2 - 6yz + \frac{14}{5} = 0 \\ x + y - 2z - 2 = 0 \end{cases}$

- eliminación de la variable x en la forma paramétrica

```
Eliminate[{2 + 3 y^2 + 6 z + 5 z^2 == 6 y (1 + z), x == 2 - y + 2 z}, x]
```

```
-2 - 6 z - 5 z^2 == 3 y^2 + y (-6 - 6 z)
```

- no es necesaria la anterior eliminación ya que en la forma implícita la curva viene dada como intersección de un plano y un cilindro al que falta la variable x con lo que sus generatrices son paralelas al eje OX y, por tanto, es el cilindro proyectante sobre el plano $x = 0$

- proyección pedida: $C_{x=0} \equiv \begin{cases} 3(y-1)^2 + 5\left(z + \frac{3}{5}\right)^2 - 6yz + \frac{14}{5} = 0 \\ x = 0 \end{cases}$

- representación gráfica: intersección entre el cilindro proyectante y el plano $x = 0$

```
prox0 = ParametricPlot3D[{0, y7[t], z7[t]}, {t, 0, 2 Pi}, PlotStyle → {Red, Thickness[0.02]}];
```

```
plano9 = ContourPlot3D[x == 0, {x, 0, 2}, {y, 0, 2}, {z, -1, 1},
   Mesh → False, ContourStyle → {LightRed, Opacity[0.4]}, AspectRatio → Automatic];
```

  - se usan las variables **g7** y **c8** de ejercicios anteriores

```
Show[c8, plano9, g7, prox0, PlotRange → All, AspectRatio → Automatic, AxesLabel → {X, Y, Z}]
```



- Solución: $\boxed{\text{Opción } c}$

Ejercicio nº10

Enunciado

Determine el paso de la hélice elíptica:

$$h(t) = \left(\cos(t),\ \frac{1}{5}\,t,\ \frac{1}{\sqrt{3}}\,\mathrm{sen}(t)\right) \quad \forall\, t \in \mathbb{R}$$

Resolución

```
Remove["Global`*"]
```

- parametrización de la hélice

```
h1 = {Cos[t], t/5, Sin[t]/Sqrt[3]};
```

- cálculo del paso

```
ps = {h1 /. t → 0, h1 /. t → 2 Pi}
```

$$\left\{\{1, 0, 0\}, \left\{1, \frac{2\pi}{5}, 0\right\}\right\}$$

```
paso = ps[[2, 2]] - ps[[1, 2]]
```

$$\frac{2\pi}{5}$$

- representación gráfica

```
cil = ContourPlot3D[x^2 + 3 z^2 == 1, {x, -1, 1}, {y, 0, 2 Pi}, {z, -1, 1}, Mesh → False, Boxed → False,
    Axes → False, ContourStyle → {LightGray, Opacity[0.4]}, AspectRatio → Automatic];

puntos = ListPointPlot3D[{ps}, PlotStyle → {Red, PointSize[Large]}];

H1 = ParametricPlot3D[h1, {t, 0, 10 Pi}, PlotStyle → {Green, Thickness[0.005]}];

Show[puntos, cil, H1, PlotRange → All, AspectRatio -> Automatic, AxesLabel → {X, Y, Z}]
```



- Solución: $\boxed{\text{Opción } \boldsymbol{f}}$